DTIC FILE COPY

AD-A214 149

TECHNICAL REPORT RD-GC-89-17

MECHANICAL EVALUATION OF THE DIFFERENTIAL DRUM CONCEPT FOR OPTICAL FIBER PAY-OUT

Stephen C. Cayson
Guidance and Control Directorate
Research, Development, and Engineering Center

JULY 1989

U.S. ARMY MISSILE COMMAND
Redstone Arsenal, Alabama 35898-5000

Approved for public release; distribution is unlimited.

SMI FORM 1021, 1 AUG 85 PREVIOUS EDITION IS OBSOLETE

89 11 06 046

DISPOSITION INSTRUCTIONS

DISCLAIMER

TRADE NAMES

TABLE OF CONTENTS

LIST OF ILLUSTRATIONS

I. INTRODUCTION

The purpose of this report is to evaluate the mechanical aspects of the differential drum concept for pulling optical fiber from a bobbin at speeds up to Mach 1. Figure 1 shows a typical differential drum fiber pay-out system schematic. As seen in the figure, two drums located on a common shaft are rotated using a prime mover. The rotating action causes optical fiber to be collected on each of the drums, therefore providing fiber pay-out from the bobbins. The large bobbin contains the fiber which is being tested and the smaller bobbin contains the fiber used to maintain signal throughput during the test. The two bobbins are connected using an optical connector as shown in the schematic. This type of arrangement allows the sensitive test equipment to remain stationary and accessible during testing and makes it easier to achieve the balancing required for the rotating system. The fiber winding is controlled using a level wind system that tracks back and forth, parallel to the shaft centerline, placing layers of fiber directly on the drum or the preceding fiber layer.

The report presented will develop the equations that govern the differential drum fiber pay-out system and then apply these equations to two different design schemes. Design considerations include the stresses and deflections present in the large drum due to centrifugal forces, the tensile stresses imposed on the fiber by the increase in diameter of the large drum, and the rotating system critical speed calculations. Also shown, for comparison only, is the maximum power required for each system assuming a constant-torque driver. The numbers for power should not be used for design purposes but should only be used as a first-cut value for the requirements. The intent of this report is to provide some insight into the mechanics of the rotating system and the problems that might occur. The power requirements are not within this scope and were, therefore, not analyzed in great detail.

II. SYSTEM ANALYSIS

For this analysis consider a differential drum system such as the one shown in Figures 2 and 3. As seen in the figure, the spokes of the large drum are tapered with the width being defined by a first-order polynomial dependent on radius. The thickness of the spokes is constant. This geometry was chosen in order to minimize the inertia of the spokes, therefore minimizing the power required for the system. It will be shown later that this geometry also provides a somewhat constant stress profile along the spoke radius. Also, the width of the hub is defined by a first-order polynomial. To maintain a constant stress in the hub it would be necessary to define the width of the hub by a higher-order polynomial, but for ease of machining a first-order, or linear, taper was chosen. The small drum consists of a rim with an inside diameter equal to the shaft and no spokes. This configuration was chosen for ease of machining and assembly and also because the shaft diameter must be made large enough in diameter to provide a high critical speed.

First of all, the inertia of the system must be found. For the rims and ends of each drum and the shaft, the inertias are found using the equations for the inertia of a right circular and are

$$I_{RL} = \pi \rho_{AL} W_D (D_{oL}^4 - D_{IL}^4)/32 \qquad (1)$$



Figure 1. Differential drum system schematic.



Figure 2. Differential drum geometry nomenclature.

$$I_{EL} = 2 \pi \rho_{AL} t_{EL} (D_{EL}^4 - D_{IL}^4)/32 \quad (2)$$

$$I_{RS} = \pi \rho_{AL} W_D (D_{os}^4 - D_{SH}^4)/32 \quad (3)$$

$$I_{ES} = 2 \pi \rho_{AL} t_{ES} (D_{ES}^4 - D_{SH}^4)/32 \quad (4)$$

$$I_{SH} = \pi \rho_{ST} L_{SH} D_{SH}^4/32 \quad (5)$$

In order to determine the inertia of the spokes and hubs of the large drum, as shown in Figure 3, consider the equation

$$I = \int R^2 \, dm \quad (6)$$

For the spokes, assuming the width is described by a first-order polynomial, Equation 6 shows that the inertia of the spokes is

$$I_{SP} = N_{SP/W} N_{W/D} \rho_{AL} t_{SL} [a (R_{IL}^4 - R_{HL}^4)/4 + b (R_{IL}^3 - R_{HL}^3)/3] \quad (7)$$

It follows that the inertia of the hubs, assuming the thickness is described by a first-order polynomial, is

$$I_H = 2 N_{W/D} \pi \rho_{AL} [a_1 (R_{HL}^5 - R_{SH}^5)/5 + b_1 (R_{HL}^4 - R_{SH}^4)/4] \quad (8)$$

The total inertia of the system is then

$$I_{TOT} = I_{RL} + I_{EL} + I_{RS} + I_{ES} + I_{SH} + I_{SP} + I_H \quad (9)$$

Knowing the inertia of the system, the power required for rotation can be found. Consider the equation

$$T_o = I_{TOT} \ddot{\theta} \quad (10)$$

which is Newton's second law of motion for a rotating system. Also consider the power equation

$$P = T_o \omega \quad (11)$$

Combining Equations 10 and 11 and assuming constant angular acceleration shows that the maximum power required is

$$P_{MAX} = I_{TOT} \omega^2/(79200 g_c t_s) \quad (12)$$

where power is measured in horsepower.

In order to calculate the radial stress and deflection of the system, the inertial forces present must be calculated. Application of Newton's second law to an element as shown in Figure 4 shows that

$$dF = (\rho \omega^2/g_c) R \, dV \quad (13)$$



Figure 3. Spoke wheel geometry.



Figure 4. <u>Differential element for inertial force determination.</u>

which is the differential inertial force created by the element. Application of Equation 13 to each of the components of the large drum results in

$$F_H = 2 \pi \rho_{AL} \omega^2 [a_1 (R_{HL}{}^4 - R^4)/4$$

$$+ b_1 (R_{HL}{}^3 - R^3)/3]/(12 g_c); \; R_{SH} < R < R_{HL} \qquad (14)$$

$$F_{SP} = N_{SP/W} \rho_{AL} t_{SL} \omega^2 [a (R_{IL}{}^3 - R^3)/3$$

$$+ b (R_{IL}{}^2 - R^2)/2]/(12 g_c); \; R_{HL} < R < R_{IL} \qquad (15)$$

$$F_R = 2 \pi \rho_{AL} \omega^2 L_{WH} (R_{OL}{}^3 - R^3)/(36 g_c)$$

$$R_{IL} < R < R_{OL} \qquad (16)$$

As an example, the development of the inertial force for the hub is shown in Appendix A. It should be pointed out that, for an element on the hub, the inertial forces acting on it are those created by the remainder of the hub plus the total inertial forces of the spokes and rim.

The equations shown above for the inertial force calculations can now be used to find the radial stress and deflection. Radial stress is found using

$$\sigma_R = F_i/A_i \qquad (17)$$

The area that the inertial force is acting on can be shown to be one of the equations below, depending on the radius, R.

$$A_H = 2 \pi R (a_1 R + b_1) \qquad (18)$$

$$R_{SH} < R < R_{HL}$$

$$A_{SP} = N_{SP/W} t_{SL} (aR + b) \qquad (19)$$

$$R_{HL} < R < R_{IL}$$

$$A_R = 2 \pi R L_{WH} \qquad (20)$$

$$R_{IL} < R < R_{OL}$$

The radial deflection can now be found using

$$\delta = \sum_{i=1}^{n} \frac{F_i \Delta R}{A_i E_{AL}} \qquad (21)$$

Equation 21 is used to find total increase in diameter of the differential drum system. The total increase in radius would be found by dividing the result of Equation 21 by two.

Next, the mass of the system must be determined. For the rims and ends of each drum and the shaft, the masses are determined using equations for the volume of right circular cylinders

$$M_{RL} = \pi \rho_{AL} W_D (R_{oL}^2 - R_{IL}^2) \qquad (22)$$

$$M_{EL} = 2 \pi \rho_{AL} t_{EL} (R_{EL}^2 - R_{IL}^2) \qquad (23)$$

$$M_{RS} = \pi \rho_{AL} W_D (R_{oS}^2 - R_{SH}^2) \qquad (24)$$

$$M_{ES} = 2 \pi \rho_{AL} t_{ES} (R_{ES}^2 - R_{SH}^2) \qquad (25)$$

$$M_{SH} = \pi \rho_{ST} L_{SH} R_{SH}^2 \qquad (26)$$

To determine the mass of the spokes and hubs of the large drum, consider the equation

$$m \int \rho \, dV \qquad (27)$$

Making the same assumptions as those stated above for the geometries of the spokes and hubs and substituting into Equation 27 shows that their masses are

$$M_{SP} = N_{SP/W} N_{W/D} \rho_{AL} t_{SL} [a (R_{IL}^2 - R_{HL}^2)/2 + b (R_{IL} - R_{HL})] \qquad (28)$$

$$M_H = 2 N_{W/D} \pi \rho_{AL} [a1 (R_{HL}^3 - R_{SH}^3)/3 + b_1 (R_{HL}^2 - R_{SH}^2)/2] \qquad (29)$$

In order to calculate the critical speed of the system, consider the rotating system shown in Figure 5. The equation which will be used to predict the critical speed is

$$N_{CRT} = \frac{387000 \, D_{SH}^2}{L_1 \, L_2} \sqrt{\frac{L_{SH}}{W}} \qquad (30)$$

where N_{CRT} is in rpm. This equation applies for a steel shaft having a modulus of elasticity of 29 Mpsi, single concentrated loads, and supported bearings. In this case, the bearings were chosen as shown in order to maximimize the critical speed. In order to account for the weight of the steel shaft, one-half of the weight of the shaft supporting the large drum will be added to the weight of the large drum to make up the variable quantity W in Equation 30.

To determine the distance between spoke wheels and the number of spokes required per wheel, consider Figure 6 which is a drawing of the cross-section of rim between either two spokes or two spoke-wheels. For the rim section between two spokes, it is assumed that the section is straight instead of curved. For a certain maximum desired deflection, y_{MAX}, it can be shown that the required beam length is



$$W = W_{LD} + W_{SH}/2$$

Figure 5. Rotating system for critical speed calculation.



Figure 6. Beam approximation for rim.

$$L_B = \left[\frac{384\ E_{AL}\ I_B\ y_{MAX}}{w}\right]^{1/4} \tag{31}$$

where w is the applied force per unit length which, in this case, represents the inertial force created by the weight of the rotating rim section. Making substitutions for the inertia and applied force per unit length into Equation 31 results in

$$L_{WH} = \left[\frac{384\ E_{AL}\ g_c\ t_R^2\ y_{MAX}}{\rho_{AL}\ R_{OL}\ \omega^2}\right]^{1/4} \tag{32}$$

which is the equation for the maximum distance required between two consecutive spoke wheels to prevent deflection greater than y_{MAX}. To determine the required angle between spoke wheels, substitute $l = R\ \theta$ into Equation 32. This results in

$$\theta_{SP} = \left[\frac{384\ E_{AL}\ g_c\ t_R^2\ y_{MAX}}{\rho_{AL}\ R_{OL}^5\ \omega^2}\right]^{1/4} \quad . \tag{33}$$

It should be noted that the deflection, y_{MAX}, is only that rim deflection relative to the spokes and does not include nor negate the radial deflection calculations shown above.

In order to determine a feasible value for y_{MAX}, two things must be considered: 1) the bending stress in the rim created by the inertial forces present and 2) the tensile stress created in the fiber by the increase in drum diameter. For the first condition, it can be shown that the maximum possible bending moment that can occur on a beam such as that one shown in Figure 6 is

$$M_{MAX} = w\ L_B^2/12 \tag{34}$$

and it occurs at the supports. The stress corresponding to Equation 34 must be solved for two cases of length; that between two spoke wheels, L_{WH}, and that between two spokes, $R\theta_{SP}$. These stresses become

$$\sigma_{B1} = 6M_{MAX}/(R\ \theta_{SP}\ t_R^2) \tag{35a}$$

$$\sigma_{B2} = 6M_{MAX}/(L_{WH}\ t_R^2) \quad . \tag{35b}$$

Now, for the second condition, Hooke's law states that strain is equal to the ratio of the change in length of a material to its initial length. It follows that tensile stress is equal to the modulus of elasticity of the material, which in this case is glass, multiplied by the strain. Combining these two statements into equation form shows, that for a certain deflection, the maximum tensile stress allowed is

$$\sigma_T = 2E_{GL}\ (y_{MAX} + \delta)/D_{OL} \tag{36}$$

where δ is the increase in radius of the rim as shown above.

To get an idea of the maximum tensile stress that can occur in the fiber without causing breakage, consider Figure 7. Assuming that the whole cross-section of fiber takes the load, the tensile stress becomes

$$\sigma_{T,MAX} = 4F_F/[\pi\ (d_B{}^2 - d_c{}^2)] \qquad (37)$$

Tensile tests have shown that the maximum allowable force on two different diameters of fiber are

$$F_F = 7.04\ lb_f \qquad d_B = 170\ \text{microns}$$

$$F_F = 15.22\ lb_f \qquad d_B = 250\ \text{microns}$$

Substitution of these values into Equation 37 shows that the maximum allowable tensile stress is

$$\sigma_T = 200\ \text{kpsi} \quad .$$

For design purposes, the maximum fiber tensile stress will be chosen as

$$\sigma_{T,MAX} = 100\ \text{kpsi} \qquad (38)$$

allowing for a Factor of Safety of 2.0.

For aluminum parts as large as the ones which will have to be used for the fabrication of the differential drum components, it is anticipated that the parts will have to be cast. The maximum yield strength for cast aluminum is 55 kpsi. For design purposes the maximum bending stress allowed will be

$$\sigma_{B,MAX} = 40\ \text{kpsi} \qquad (39)$$

which indicates a Factor of Safety of 1.4.

The final trade-off to be considered is the amount of fiber payed-out during the time it takes to reach full speed. Using the assumption made above that the drum will speed up at constant angular acceleration, application of basic kinematic equations show that

$$l_F = R_{OL}\ \omega\ t_s/2 \qquad (40)$$

where t_s is the time allowed to reach full pay-out speed.



Figure 7. Optical fiber.

III. PAY-OUT MACHINE DESIGN

A computer program has been written using FORTRAN which will calculate the weight, critical speed, maximum power and torque required, fiber payed-out during speed-up time, and the radial deflection and stress of a differential drum system given the geometry. It also generates data files containing the radial deflection and stress as a function of radius which can be used to make plots. This program is shown in Appendix B. Two differential drum system designs will be considered in this report: 1) one with a large drum outside diameter of 25 inches; and 2) one with a large drum outside diameter of 52 inches. For both cases the small drum has an outside diameter of 5.2 inches. Considering the ratio of diameters for each differential drum design shows that the amount of extra fiber needed for the optical link using the 25 inch system is 4.58 km and 2.20 km for the 52 inch system, assuming test bobbin fiber lengths of 22 km. The corresponding optical link pay-out velocities are Mach 0.21 for the 25 inch system and Mach 0.10 for the 52 inch system. The analyses are shown in the following sections.

A. Design of a Twenty-five Inch Diameter Pay-Out Machine

First of all, for the 25 inch drum design, the drum width, W_D, must be chosen. This value is directly related to how many layers of fiber will accumulate on the drums during pay-out. The number of layers, assuming the fiber rows are wrapped side-by-side, can be found using

$$N_L = L_{F,B}\, d_B/(\pi\, D_{OL}\, W_D) \qquad (41)$$

For a 22 km test bobbin of 250 micron diameter fiber being wound on a 25 inch diameter, 8 inch wide drum, the number of layers accumulated will be 13.6. The number of layers will be 9.2 for the 170 micron diameter fiber. These numbers apply for both the large and small drum.

Now Equations 32 and 33 can be used to find the maximum distance between spoke wheels and the maximum angle between spokes. For a relative deflection, y_{MAX}, of 0.005 inches, a rim thickness, t_R, of 0.5 inches, and an angular velocity of 1075.2 rad/sec (corresponds to a Mach 1.0 pay-out velocity) the maximums are

$$L_{WH,MAX} = 3.23 \text{ in} \qquad \theta_{SP,MAX} = 14.8 \text{ deg} \quad .$$

For this case it was decided to make

$$L_{WH} = 2.33 \text{ in} \qquad \theta_{SP} = 12.0 \text{ deg}$$

which implies that there are four spokes wheels, each having 30 spokes. Substituting these values back into Equations 32 and 33 show that the actual deflections are then

$$y_{MAX,L} = 0.0014 \text{ in} \qquad y_{MAX,} = 0.0022 \text{ in}$$

which are both less than the 0.005 inches specified as a maximum above.

Several preliminary designs were investigated before deciding on the one with the parameters shown in Table I. This design was chosen after iterating through the equations discussed above to find suitable values for the stresses in the large drum and the fiber. As seen in the table and as discussed on previous page, this differential drum system has two 8.0 inch wide drums, the larger one having four spoke wheels such as those shown in Figure 3. The geometry of the spoke was determined to allow a maximum radial stress not exceeding the value stated in Section II for cast aluminum. The geometry describing the system was included as input to the program shown in Appendix B and the results of execution are shown in Table I. As seen in the results, the maximum radial stress occurs at a pay-out velocity of Mach 1.0 and is 39 kpsi, which is below the design stress of 40 kpsi. Figures 8 and 9 shows the radial stress deflection plots as a function of radius for the case when the pay-out velocity is Mach 0.7.

Now a check should be done to verify that the tensile stress in the fiber and the bending stress on the large drum rim are within the limits stated above. Referring to the program output in Table I shows that the maximum radial deflection, δ, is 0.01651 inches. Taking this value and the maximum of the two y_{MAX}'s shown above (0.0022 in) and substituting into Equation 36 show that the tensile stress in the fiber due to elongation is

$$\sigma_T = 10.0 \text{ kpsi}$$

which is well below the maximum allowable tensile stress of 100 kpsi.

To find the bending stress on the rim substitute the proper values into Equations 34, 35a, and 35b. This shows that the maximum bending stress occurs between two spokes and is

$$\sigma_{B,MAX} = 25.7 \text{ kpsi}$$

which is below the maximum allowable bending stress of cast aluminum of 40 kpsi. These calculations are shown in Appendix C.

Also shown in Table I is the system critical speed, which is 48000 rpm. This value is over four times greater than the 10000 rpm required for pay-out at Mach 1.0 and is considered to be a safe margin to prevent any system vibration from occurring. Finally, the maximum torque and power required and the amount of fiber payed-out during speed-up is shown for each case of desired pay-out speed and speed-up time.

B. Design of a Fifty-Two Inch Diameter Pay-Out Machine

Techniques similar to those described above for the 25 inch diameter pay-out machine were used to design the 52 inch diameter version, whose geometry variables are shown in Table II. First of all the drum width was decided to be 4.0 inches, which, using Equation 41, results in 13.0 layers of 250 micron diameter fiber and 8.9 layers of 170 micron diameter fiber being wound onto the drums using a 22 km long test bobbin.

TABLE I. Twenty-Five Inch Diameter Drum Program Output

DIFFERENTIAL DRUM
FIBER PAY-OUT DEVICE

SYSTEM GEOMETRY

	LARGE DRUM	SMALL DRUM
RIM OUTSIDE DIAMETER (in)	25.0	5.2
RIM INSIDE DIAMETER (in)	24.0	3.5
HUB OUTSIDE DIAMETER (in)	10.84	NO HUB
HUB THICKNESS AT SHAFT (in)	1.50	NO HUB
HUB THICKNESS AT SPOKE (in)	.50	NO HUB
SPOKE WIDTH AT HUB (in)	1.10	NO SPOKES
SPOKE WIDTH AT RIM (in)	.55	NO SPOKES
SPOKE THICKNESS (in)	.5	NO SPOKES
# OF SPOKES PER WHEEL	30.	NO SPOKES
# OF WHEELS PER DRUM	4.	NO WHEELS
END OUTSIDE DIAMETER (in)	26.0	6.2
END THICKNESS (in)	.5	.5
SHAFT DIAMETER (in)	3.5	3.5
SHAFT LENGTH (in)	11.0	19.0
DRUM LENGTH (in)	8.0	8.0

WEIGHT OF LARGE DRUM = 99.4 lbf (ALUMINUM)
WEIGHT OF SMALL DRUM = 11.1 lbf (ALUMINUM)
WEIGHT OF SHAFT = 82.0 lbf (STEEL)

SYSTEM CRITICAL SPEED = 48584.3 RPM

MAXIMUM POWER AND TORQUE REQUIRED
AND AMOUNT OF FIBER PAYED-OUT DURING SPEED-UP

	PAYOUT VELOCITY		
	MACH .5 (5133.7 RPM)	MACH .7 (7187.2 RPM)	MACH 1.0 (10267.4 RPM)
SPEED-UP TIME (sec)			
3.0	327.6 HP 4021.3 in-lbf .26 km	642.0 HP 5629.9 in-lbf .36 km	1310.2 HP 8042.7 in-lbf .51 km
5.0	196.5 HP 2412.8 in-lbf .43 km	385.2 HP 3377.9 in-lbf .60 km	786.1 HP 4825.6 in-lbf .85 km
10.0	98.3 HP 1206.4 in-lbf .85 km	192.6 HP 1689.0 in-lbf 1.19 km	393.1 HP 2412.8 in-lbf 1.71 km
RADIAL DEFLECTION	.00413 in	.00809 in	.01651 in
MAX RADIAL STRESS	9763. psi	19135. psi	39052. psi

TABLE II. Fifty-Two Inch Diameter Drum Program Output

DIFFERENTIAL DRUM
FIBER PAY-OUT DEVICE

SYSTEM GEOMETRY

	LARGE DRUM	SMALL DRUM
RIM OUTSIDE DIAMETER (in)	52.0	5.2
RIM INSIDE DIAMETER (in)	51.0	3.5
HUB OUTSIDE DIAMETER (in)	10.84	NO HUB
HUB THICKNESS AT SHAFT (in)	1.50	NO HUB
HUB THICKNESS AT SPOKE (in)	.50	NO HUB
SPOKE WIDTH AT HUB (in)	.90	NO SPOKES
SPOKE WIDTH AT RIM (in)	.50	NO SPOKES
SPOKE THICKNESS (in)	.5	NO SPOKES
* OF SPOKES PER WHEEL	36.	NO SPOKES
* OF WHEELS PER DRUM	3.	NO WHEELS
END OUTSIDE DIAMETER (in)	53.0	6.2
END THICKNESS (in)	.5	.5
SHAFT DIAMETER (in)	3.5	3.5
SHAFT LENGTH (in)	7.0	15.0
DRUM LENGTH (in)	''4.0	4.0

WEIGHT OF LARGE DRUM = 144.3 lbf (ALUMINUM)
WEIGHT OF SMALL DRUM = 6.6 lbf (ALUMINUM)
WEIGHT OF SHAFT = 60.1 lbf (STEEL)

SYSTEM CRITICAL SPEED = 82535.6 RPM

MAXIMUM POWER AND TORQUE REQUIRED AND AMOUNT OF FIBER PAYED-OUT DURING SPEED-UP

	PAYOUT VELOCITY		
SPEED-UP TIME (sec)	MACH .5 (2468.1 RPM)	MACH .7 (3455.4 RPM)	MACH 1.0 (4936.3 RPM)
3.0	440.2 HP 11240.1 in-lbf .26 km	862.7 HP 15736.1 in-lbf .36 km	1760.7 HP 22480.2 in-lbf .51 km
5.0	264.1 HP 6744.1 in-lbf .43 km	517.6 HP 9441.7 in-lbf .60 km	1056.4 HP 13488.1 in-lbf .85 km
10.0	132.1 HP 3372.0 in-lbf .85 km	258.8 HP 4720.8 in-lbf 1.19 km	528.2 HP 6744.1 in-lbf 1.71 km
RADIAL DEFLECTION	.00963 in	.01888 in	.03852 in
MAX RADIAL STRESS	8571. psi	16799. psi	34284. psi



Figure 8. <u>Stress distribution for 25 inch drum Mach 0.7 pay-out velocity.</u>



Figure 9. Radial deflection for 25 inch drum Mach 0.7 pay-out velocity.

Evaluation of Equations 32 and 33 show that for a relative maximum deflection of 0.003 inches, the maximum distance and angle between spokes is

$$L_{WH,MAX} = 3.43 \text{ in} \qquad \theta_{SP,MAX} = 7.7 \text{ deg} .$$

The relative deflection was chosen as 0.003 inches to allow for the increase in radial deflection that will occur with the large diameter drum and also the increase in inertial force of the rim section. For the design, the following values were selected for L_{WH} and θ_{SP}

$$L_{WH} = 2.0 \text{ in} \qquad \theta_{SP} = 10.0 \text{ deg}$$

which implies that there are three spokes wheels containing 36 spokes each. The actual deflections then become

$$y_{MAX,L} = 0.0004 \text{ in} \qquad y_{MAX,} \quad = 0.0085 \text{ in} \quad .$$

The deflection between two spokes, $y_{MAX,}$, is larger than the 0.003 inches maximum stated above but it will be shown that this deflection can be allowed without sacrificing the system integrity.

Next, the bending stresses in the rim section must be calculated. Using Equations 34, 35a, and 35b show that the maximum bending stress occurs in the plane between two spokes and is

$$\sigma_{B,MAX} = 36.6 \text{ kpsi}$$

which is below the design stress of cast aluminum chosen as 40 kpsi. To calculate the tensile stress on the optical fiber, the computer program must be executed to determine the maximum radial deflection. The results of program execution are shown in Table II and they show that the radial deflection is 0.03852 inches for a pay-out velocity of Mach 1.0. Substituting this value into Equation 36 along with the value for $y_{MAX,}$, which is 0.0085 inches, shows that the tensile stress in the fiber due to the increase in its length is

$$\sigma_T = 12.1 \text{ kpsi}$$

which is significantly lower than the design tensile stress of 100 kpsi. The maximum radial stress on the drum due to inertial forces is also shown in TableII as calculated in the computer program. This value, occurring at a pay-out velcoity of Mach 1.0, is 34.3 kpsi and is within the design constraint of 40 kpsi. Figures 10 and 11 show plots of the radial stress and deflection versus radius for the case of Mach 0.7 pay-out velocity.

Also shown in the results is the system critical speed, 83000 rpm, which is over 15 times greater than the drum speed required for fiber pay-out at Mach 1.0 of 5000 rpm. This margin is considered safe to prevent system vibration. Finally, the maximum torque and power required and the amount of fiber payed-out during speed-up is shown for each case of desired pay-out speed and speed-up time.



Figure 10. <u>Stress distribution for 52 inch drum Mach 0.7 pay-out velocity.</u>



Figure 11. Radial deflection for 52 inch drum Mach 0.7 pay-out velocity.

IV. CONCLUSIONS AND RECOMMENDATIONS

Two designs that provide optical fiber pay-out using the differential drum concept have been presented. This concept gathers the fiber in a systematic way and allows continuous signal throughput transmission during testing by means of an optical link bobbin that is connected to the test bobbin and payed-out simultaneously. This necessity of a smaller optical link bobbin could be deemed one of the primary disadvantages of the differential drum concept. One reason is the added cost of the optical link bobbin which is required for each test and the other is the added signal attenuation that would be present in the optical link bobbin. Another main disadvantage of these systems is the amount of power required to reach the desired fiber pay-out velocity. Tables I and II show these results for various pay-out speeds and speed-up times. Comparison of the power required for the two designs shows, of course, that the system with the 25 inch diameter requires less power. Further investigation of the power requirements for 25 inch system, shown in Table I, shows that, for the same size engine, higher pay-out speeds can be attained using longer speed-up times. Although the amount of fiber payed-out during speed-up would increase, the results in Table I shows that this value does not get very large. The advantages of this concept are that the mechanics can be accomplished without exotic materials or new technology and that signal throughput is maintained during pay-out. Cast aluminum has been used in the design and has been shown to provide sufficient strength capabilities for this application, but it is possible that a plate aluminum of higher strength can be used if available. Another advantage of this concept is the low tensile stress imparted to the fiber due to the mechanics of the system, i.e., the radial deflection of the take-up drum. This low stress allows for stresses induced by other means, such as microbending and discontinuities in winding the fiber, to occur and possibly, depending on the severity of the phenomena, not reach the failure point of the fiber.

REFERENCES

1. Joseph Edward Shigley, and Larry D. Mitchell, Mechanical Engineering Design, 4th ed, McGraw-Hill Book Company, Copyright 1983.

2. Ferdinand P. Beer, and E. Russell Johnson, Jr., Mechanics of Materials, McGraw-Hill Book Company, Copyright 1981.

3. Erik Oberg, Franklin D. Jones, and Holbrook L. Horton, Machinery's Handbook, 22nd ed, Industrial Press Inc., Copyright 1984.

4. Material Selector 1989 provided by Materials Engineering Magazine, Penton Publishing Company.

LIST OF NOMENCLATURE

I_{RL} = Inertia of rim of large drum (in^2-lb_m)

I_{EL} = Inertia of ends of large drum (in^2-lb_m)

I_{RS} = Inertia of rim of small drum (in^2-lb_m)

I_{ES} = Inertia of ends of small drum (in^2-lb_m)

I_{SH} = Inertia of shaft (in^2-lb_m)

I_{SP} = Inertia of spokes of large drum (in^2-lb_m)

I_H = Inertia of hubs of large drum (in^2-lb_m)

I_{TOT} = Total inertia of system (in^2-lb_m)

ρ_{AL} = Density of aluminum (lb_m/in^3)

ρ_{ST} = Density of steel (lb_m/in^3)

D_{oL} = Outside diameter of large drum (in)

D_{IL} = Inside diameter of large drum (in)

D_{EL} = Outside diameter of large drum ends (in)

D_{oS} = Outside diameter of small drum (in)

D_{ES} = Outside diamter of small drum ends (in)

D_{HL} = Outside diamter of large drum hubs (in)

D_{SH} = Shaft diamter (in)

R_{oL} = Outside radius of large drum rim [$D_{oL}/2$] (in)

R_{IL} = Inside radius of large drum rim [$D_{IL}/2$] (in)

R_{HL} = Outside radius of large drum hubs [$D_{HL}/2$] (in)

L_{SH} = Shaft length (in)

W_D = Drum width (in)

t_{EL} = End thickness of large drum (in)

t_{ES} = End thickness of small drum (in)

t_{SL} = Spoke thickness of large drum (in)

a, b = Coefficients describing geometry of spokes (in)

LIST OF NOMENCLATURE (Cont'd)

a_1, b_1 = Coefficients describing geometry of hubs (in)

$N_{SP/W}$ = Number of spokes per wheel

$N_{W/D}$ = Number of wheels per drum

T_o = Torque required by system (in-lb_f)

$\ddot{\theta}$ = Angular acceleration (rad/sec^2)

P = Power required by system (HP)

ω = Angular velocity (rad/sec)

g_c = Gravitational constant (lb_m/slug)

t_s = Speed-up time (sec)

F_H = Total inertial force of hub (lbf)

F_{SP} = Total inertial force of spokes (lb_f)

F_R = Total inertial force of large drum rim (lb_f)

σ_R = Radial stress on element (psi)

F_i = Inertial force acting on element (lb_f)

A_i = Area of element (in^2)

A_H = Area of hub element (in^2)

A_{SP} = Area of spoke element (in^2)

A_R = Area of rim element (in^2)

δ = Total radial deflection (in)

ΔR = Integration step size for deflection calculation (in)

E_{AL} = Modulus of elasiticity of aluminum (psi)

M_{RL} = Mass of large drum rim (lb_m)

M_{EL} = Mass of ends of large drum (lb_m)

M_{RS} = Mass of small drum rim (lb_m)

M_{ES} = Mass of ends of small drum (lb_m)

M_{SH} = Mass of shaft (lb_m)

LIST OF NOMENCLATURE (Cont'd)

m_{SP} = Mass of spokes (lb_m)

m_H = Mass of hubs (lb_m)

N_{CRT} = Critical speed of system (rpm)

L_1 = Distance from left bearing to large drum CL (in)

L_2 = Distance from large drum CL to right bearing (in)

W = Weight used in critical speed calculation (lb_f)

L_B = Beam length used for stress calculations (in)

I_B = Inertia of cross-section of rim (in^4)

y_{MAX} = Maximum rlative deflection of rim (in)

w = Distributed force over rim section (lb_f/in)

L_{WH} = Distance between two spoke wheels (in)

θ_{SP} = Angle betweent two spokes (rad)

t_R = Thickness of large drum rim (in)

M_{MAX} = Maximum bending moment on rim section (in-lb_f)

σ_B = Maximum bending stress on rim section (psi)

σ_T = Tensile stress on fiber due to elongation (psi)

E_{GL} = Modulus of elasticity of glass (psi)

F_F = Breaking force on fiber (lb_f)

d_B = Buffer diameter of fiber (in)

d_c = Core diameter of fiber (in)

l_F = Amount of fiber payed-out during speed-up (km)

N_L = Number of layers of fiber wound onto drum

$L_{F,B}$ = Length of fiber on test bobbin (km)

APPENDIX A

Determine the inertial force as a function of radius for the hub.



Let $W_H(R)$ be a first-order polynomial of the form

$$W_H(R) = a_1 R + b_1 \quad .$$

Apply Newton's law to an element

$$dF = \frac{a\ dm}{g_c} = \frac{V^2}{R}\ \frac{\rho dV}{g_c} \qquad V = R\omega$$

$$dF = \frac{\rho\omega^2}{g_c}\ RdV \qquad dV = 2\pi\ W_H(R)dR$$

$$dF = \frac{2\pi\rho\omega^2}{g_c}\ RW_H(R)dR$$

$$\int_0^F dF = \frac{2\pi\rho\omega^2}{g_c} \int_R^{R_{HL}} (a_1R^2 + b_1R)\ dR$$

$$F(R) = \frac{2\pi\rho\omega^2}{g_c} \left[\frac{a_1\ R^3}{3} + \frac{b_1\ R^2}{2} \right]_R^{R_{HL}}$$

$$F(R) = \frac{2\pi\rho\omega^2}{g_c} \left[\frac{a_1}{3} (R_{HL}{}^3 - R^3) + \frac{b_1}{2} (R_{HL}{}^2 - R^2) \right]$$

$$F(R) = \frac{2\pi\rho w^2}{12\ g_c}\left[\frac{a_1}{3}(R_{HL}{}^3 - R^3) + \frac{b_1}{2}(R_{HL}{}^2 - R^2)\right][lb_f]$$

APPENDIX B
PROGRAM DIFFERENTIAL DRUM

```
      PROGRAM DIFFERENTIAL DRUM

      REAL L,LDI,LSH,LW,MACH,MXSTRS

      DIMENSION RUNUP(3),MACH(3),OMEGA(3),RPM(3),DDEFI(3),RDEFI(3),
     &          DEF(3),RADIAL(3),MXSTRS(3),TORQUE(3,3),POWER(3,3),
     &          FIBER(3,3)

      OPEN(4,FILE='LPT1')
      OPEN(5,FILE='MACH5.DAT',STATUS='NEW')
      OPEN(6,FILE='MACH7.DAT',STATUS='NEW')
      OPEN(7,FILE='MACH1.DAT',STATUS='NEW')

C                   *********************************
C                   *****  PROGRAM DESCRIPTION  *****
C                   *********************************
C        This program will calculate several design parameters for the
C   differential drum concept for optical fiber pay-out.  After the
C   required system geometry is provided as input by the user and the
C   program is executed, the weight of each component of the system is
C   calculated.  These calculations assume that the shaft is made of
C   steel and all other parts are made of aluminum.  The inertia of
C   each component is then calculated followed by the system critical
C   speed.  At this point in the program, a loop which is dependent
C   on the desired pay-out Mach Number begins.  This loop calculates
C   the radial stress and deflection of the large drum for each
C   pay-out speed.  Within this loop is another that is dependent on
C   the desired time to reach final pay-out speed.  This loop calculates
C   the maximum torque and power required and the amount of fiber
C   payed-out during speed-up.
C
C                  *******************************
C                  *****  PROGRAM VARIABLES  *****
C                  *******************************
C   Any varible with the character, -, denotes that either an L or S
C   is substituted in it's place.  This denotes a varible that applies
C   to either the large or small drum.
C
C   VARIABLE                   DESCRIPTION
C         *****  INPUTS  *****
C   DO-. . . . . . . . . . . . RIM OUTSIDE DIAMETER (in)
C   DI-. . . . . . . . . . . . RIM INSIDE DIAMETER (in)
C   T- . . . . . . . . . . . . RIM THICKNESS (in)
C   L. . . . . . . . . . . . . LENGTH OF DRUMS (in)
C   LW . . . . . . . . . . . . DISTANCE BETWEEN SPOKE WHEELS (in)
C   TS-. . . . . . . . . . . . THICKNESS OF SPOKES (in)
C   WSPH . . . . . . . . . . . WIDTH OF SPOKE AT HUB CONNECTION (in)
C   WSPR . . . . . . . . . . . WIDTH OF SPOKE AT RIM CONNECTION (in)
C   SPOKE- . . . . . . . . . . NUMBER OF SPOKES PER WHEEL
C   WHEEL- . . . . . . . . . . NUMBER OF SPOKE-WHEELS PER DRUM
C   DH-. . . . . . . . . . . . HUB OUTSIDE DIAMETER (in)
C   THL1 . . . . . . . . . . . LARGE HUB WIDTH AT SHAFT CONNECTION (in)
C   TE-. . . . . . . . . . . . THICKNESS OF ENDS (in)
C   DE-. . . . . . . . . . . . END OUTSIDE DIAMETER (in)
C   DSH. . . . . . . . . . . . SHAFT DIAMETER FOR SMALL DRUM (in)
```

```
C  LSH. . . . . . . . . . . SHAFT LENGTH FOR SMALL DRUM (in)
C  CRTD . . . . . . . . . . SHAFT DIAMETER FOR LARGE DRUM (in)
C  CRTL . . . . . . . . . . SHAFT LENGTH FOR LARGE DRUM (in)
C  RHOAL. . . . . . . . . . DENSITY OF ALUMINUM (lbm/in^3)
C  RHOST. . . . . . . . . . DENSITY OF STEEL (lbm/in^3)
C  E. . . . . . . . . . . . MODULUS OF ELASTICITY OF ALUMINUM (psi)
C  RUNUP. . . . . . . . . . TIME TO REACH DESIRED SPEED (sec)
C  MACH . . . . . . . . . . DESIRED FIBER SPEED (Mach Number)
C  DELR . . . . . . . . . . INTEGRATION STEP SIZE (in)
C  GC . . . . . . . . . . . GRAVITATIONAL CONSTANT (lbm/slug)

C        *****  OUTPUTS  *****
C  D1 . . . . . . . . . . . DISTANCE FROM LEFT BEARING TO DRUM CL (in)
C  D2 . . . . . . . . . . . DISTANCE FROM DRUM CL TO RIGHT BEARING (in)
C  A,B. . . . . . . . . . . COEFFICIENTS OF POLYNOMIAL DESCRIBING
C                           WIDTH OF SPOKE OF LARGE DRUM
C  A1, B1 . . . . . . . . . COEFFICIENTS OF POLYNOMIAL DESCRIBING
C                           THICKNESS OF HUB OF LARGE DRUM
C  OMEGA. . . . . . . . . . ANGULAR VELOCITY OF SYSTEM (rad/sec)
C  RPM  . . . . . . . . . . SPEED OF SYSTEM (rpm)
C  W-R. . . . . . . . . . . WEIGHT OF RIM (lbf)
C  W-S. . . . . . . . . . . TOTAL WEIGHT OF SPOKES (lbf)
C  W-H. . . . . . . . . . . TOTAL WEIGHT OF HUBS (lbf)
C  W-E. . . . . . . . . . . TOTAL WEIGHT OF ENDS (lbf)
C  W-F. . . . . . . . . . . TOTAL WEIGHT OF FLANGES (lbf)
C  W-DTOT . . . . . . . . . TOTAL WEIGHT OF DRUM (lbf)
C  WSHAFT . . . . . . . . . WEIGHT OF SHAFT (lbf)
C  WSCRT. . . . . . . . . . LARGE DRUM SHAFT WEIGHT (lbf)
C  R-I. . . . . . . . . . . INERTIA OF RIM (in^2-lbm)
C  S-I. . . . . . . . . . . TOTAL INERTIA OF SPOKES (in^2-lbm)
C  H-I. . . . . . . . . . . TOTAL INERTIA OF HUBS (in^2-lbm)
C  E-I. . . . . . . . . . . TOTAL INERTIA OF ENDS (in^2-lbm)
C  -DI. . . . . . . . . . . TOTAL INERTIA OF DRUM (in^2-lbm)
C  SHI. . . . . . . . . . . INERTIA OF SHAFT (in^2-lbm)
C  TOTI . . . . . . . . . . TOTAL INERTIA OF SYSTEM (in^2-lbm)
C  RPMCRT . . . . . . . . . CRITICAL SPEED (rpm)
C  FRIM . . . . . . . . . . TOTAL INERTIAL FORCE OF RIM (lbf)
C  FSPOKE . . . . . . . . . TOTAL INERTIAL FORCE OF SPOKES (lbf)
C  FHUB . . . . . . . . . . TOTAL INERTIAL FORCE OF HUB (lbf)
C  FI . . . . . . . . . . . INERTIAL FORCE ACTING ON ELEMENT (lbf)
C  AI . . . . . . . . . . . AREA OF ELEMENT (in^2)
C  DDEFI. . . . . . . . . . DIAMETRAL DEFLECTION OF ELEMENT (in)
C  RDEFI. . . . . . . . . . RADIAL DEFLECTION OF ELEMENT (in)
C  DEF. . . . . . . . . . . TOTAL RADIAL DEFLECTION OF SYSTEM (in)
C  RADIAL . . . . . . . . . RADIAL STRESS ON ELEMENT (psi)
C  MXSTRS . . . . . . . . . MAXIMUM RADIAL STRESS OF SYSTEM (psi)
C  TORQUE . . . . . . . . . MAXIMUM TORQUE REQUIRED (in-lbf)
C  POWER. . . . . . . . . . MAXIMUM POWER REQUIRED (hp)
C  FIBER. . . . . . . . . . FIBER USED DURING SPEED-UP (km)

C  ******************************
C  *****  SYSTEM CONSTANTS  *****
C  ******************************
C -- GEOMETRY (all dimensions in inches)
C -- SHAFT
```

```
      DSH = 3.5
      RSH = DSH/2
      LSH = 15.0

C -- DISTANCES FOR CRITICAL FREQUENCY CALCULATIONS
      CRTD = 3.5
      CRTL = 7.0
      D1 = CRTL/2
      D2 = CRTL/2

C -- LARGE DRUM - RIM
      DOL = 52.
      TL = .5
      DIL = DOL - 2*TL
      ROL = DOL/2
      RIL = DIL/2
      L = 4.
      LW = 2.

C -- LARGE DRUM - SPOKES
      TSL = .5
      WHEELL = 3.
      SPOKEL = 36.

C -- LARGE DRUM - HUB
      DHL = 10.84
      RHL = DHL/2

C   COEFFICIENTS THAT DESCRIBE WIDTH OF SPOKES
C   A IS THE COEFFICIENT THAT IS MULTIPLIED BY THE RADIUS
      WSPH = 0.9
      WSPR = 0.5
      A = (WSPH-WSPR)/(RHL-RIL)
      B = WSPH - A*RHL

C   COEFFICIENTS THAT DESCRIBE THICKNESS OF HUBS
C   A1 IS THE COEFFICIENT THAT IS MULTIPLIED BY THE RADIUS
      THL1 = 1.5
      A1 = (THL1-TSL)/(RSH-RHL)
      B1 = THL1 - A1*RSH
      THSH = A1*RSH + B1
      THSP = A1*RHL + B1

C -- LARGE DRUM - ENDS
      TEL = .5
      DEL = 53.
      REL = DEL/2

C -- SMALL DRUM - RIM
      DOS = 5.2
      DIS = DSH
      ROS = DOS/2
      RIS = DIS/2

C -- SMALL DRUM - ENDS
```

```
      TES = .5
      DES = 6.2
      RES = DES/2

C -- MATERIAL CONSTANTS
      RHOAL = .098
      RHOST = .284
      E = 10000000.

      DELR = .01
      PI = 3.14159
      GC = 32.174

C -- SPEED-UP TIMES (sec)
      RUNUP(1) = 3.0
      RUNUP(2) = 5.0
      RUNUP(3) = 10.0

C -- PAYOUT SPEEDS (Mach Number)
      MACH(1) = 0.5
      MACH(2) = 0.7
      MACH(3) = 1.0

C -- CALCULATE ANGULAR VELOCITIES (rad/sec)
      DO 500 I = 1,3
      OMEGA(I) = MACH(I)*1120*12/(DOL/2)
 500  RPM(I) = OMEGA(I)*30/PI

C   ****************************************
C   *****  PRINT SYSTEM CHARACTERISTICS  *****
C   ****************************************
      WRITE(4,100)
 100  FORMAT(
     & /////33X,'DIFFERENTIAL DRUM'
     &,/32X,'FIBER PAY-OUT DEVICE',//)

      WRITE(4,110)
 110  FORMAT(
     & 34X,'SYSTEM GEOMETRY'
     &,//41X,'LARGE',20X,'SMALL'
     &,/41X,'DRUM',21X,'DRUM')

      WRITE(4,120)DOL,DOS
 120  FORMAT(7X,'RIM OUTSIDE DIAMETER (in)',9X,F4.1,21X,F4.1)

      WRITE(4,130)DIL,DIS
 130  FORMAT(7X,'RIM INSIDE DIAMETER (in)',10X,F4.1,21X,F4.1)

      WRITE(4,140)DHL
 140  FORMAT(7X,'HUB OUTSIDE DIAMETER (in)',8X,F5.2,19X,'NO HUB')

      WRITE(4,150) THSH,THSP
 150  FORMAT(
     & 7X,'HUB THICKNESS AT SHAFT (in)',7X,F4.2,19X,'NO HUB',
     &/7X,'HUB THICKNESS AT SPOKE (in)',7X,F4.2,19X,'NO HUB')
```

```
      WRITE(4,160) WSPH,WSPR
 160  FORMAT(
     & 7X,'SPOKE WIDTH AT HUB (in)',11X,F4.2,16X,'NO SPOKES',
     &/7X,'SPOKE WIDTH AT RIM (in)',11X,F4.2,16X,'NO SPOKES')

      WRITE(4,170)TSL
 170  FORMAT(7X,'SPOKE THICKNESS (in)',15X,F3.1,16X,'NO SPOKES')

      WRITE(4,180)SPOKEL
 180  FORMAT(7X,'# OF SPOKES PER WHEEL',14X,F3.0,16X,'NO SPOKES')

      WRITE(4,190)WHEELL
 190  FORMAT(7X,'# OF WHEELS PER DRUM',15X,F3.0,16X,'NO WHEELS')

      WRITE(4,200)DEL,DES
 200  FORMAT(7X,'END OUTSIDE DIAMETER (in)',9X,F4.1,21X,F4.1)

      WRITE(4,210)TEL,TES
 210  FORMAT(7X,'END THICKNESS (in)',17X,F3.1,22X,F3.1)

      WRITE(4,220)CRTD,DSH,CRTL,LSH,L,L
 220  FORMAT(
     & 7X,'SHAFT DIAMETER (in)',16X,F3.1,22X,F3.1,
     &/7X,'SHAFT LENGTH (in)',17X,F4.1,21X,F4.1
     &/7X,'DRUM LENGTH (in)',18X,F4.1,21X,F4.1)

C    ****************************************
C    *****  CALCULATE WEIGHT OF SYSTEM  *****
C    ****************************************
C -- WEIGHT OF LARGE DRUM COMPONENTS
C  RIM
      WLR = RHOAL*PI*L*(ROL**2 - RIL**2)
C  SPOKES
      WLS = SPOKEL*WHEELL*RHOAL*TSL*(A*(RIL**2 - RHL**2)/2
     &    + B*(RIL - RHL))
C  HUBS
      WLH = WHEELL*2*PI*RHOAL*(A1/3*(RHL**3-RSH**3)+
     &      B1/2*(RHL**2-RSH**2))
C  ENDS
      WLE = 2*RHOAL*PI*TEL*(REL**2 - RIL**2)

      WLDTOT = WLR+WLS+WLH+WLE

C -- WEIGHT OF SMALL DRUM COMPONENTS
C  RIM
      WSR = RHOAL*PI*L*(ROS**2 - RIS**2)
C  ENDS
      WSE = 2*RHOAL*PI*TES*(RES**2 - RSH**2)

      WSDTOT = WSR+WSS+WSH+WSE

C -- WEIGHT OF SHAFT
      WSHAFT = RHOST*PI*(RSH**2*LSH + (CRTD/2)**2*CRTL)
      WSCRT = RHOST*PI*(CRTD/2)**2*CRTL
```

```
C   ****************************
C   *****  PRINT WEIGHTS  *****
C   ****************************
      WRITE(4,230)WLDTOT,WSDTOT,WSHAFT
 230  FORMAT(
     &/7X,'WEIGHT OF LARGE DRUM          = ',F5.1,' lbf (ALUMINUM)',
     &/7X,'WEIGHT OF SMALL DRUM          = ',F5.1,' lbf (ALUMINUM)',
     &/7X,'WEIGHT OF SHAFT               = ',F5.1,' lbf (STEEL)')

C   ******************************************
C   *****  CALCULATE INERTIA OF SYSTEM  *****
C   ******************************************
C -- INERTIA OF LARGE DRUM COMPONENTS
C  RIM
      RLI = PI*RHOAL*L*(DOL**4 - DIL**4)/32
C  SPOKES
      SLI = SPOKEL*WHEELL*RHOAL*TSL*(A*(RIL**4 - RHL**4)/4 +
     &   B*(RIL**3 - RHL**3)/3)
C  HUBS
      HLI = WHEELL*2*PI*RHOAL*(A1/5*(RHL**5-RSH**5)+
     &       B1/4*(RHL**4-RSH**4))
C  ENDS
      ELI = 2*PI*RHOAL*TEL*(DEL**4 - DIL**4)/32

      LDI = RLI+SLI+HLI+ELI

C -- INERTIA OF SMALL DRUM COMPONENTS
C  RIM
      RSI = PI*RHOAL*L*(DOS**4 - DIS**4)/32
C  ENDS
      ESI = 2*PI*RHOAL*TES*(DES**4 - DSH**4)/32

      SDI = RSI+SSI+HSI+ESI+FSI

C -- INERTIA OF SHAFT
      SHI = PI*RHOST/32*(LSH*DSH**4 + CRTL*CRTD**4)

      TOTI = LDI+SDI+SHI

C   ***************************************************
C   *****  CRITICAL FREQUENCY OF ROTATING SYSTEM  *****
C   ***************************************************

      RPMCRT = 387000*CRTD**2*SQRT(CRTL/(WLDTOT+WSCRT/2))/(D1*D2)

C   ***********************************************
C   *****  BEGIN MACH NUMBER-DEPENDANT LOOP  *****
C   ***********************************************

      DO 510 I=1,3
      K = I + 4
C   ****************************************************
C   *****  CALCULATE RADIAL DEFLECTION AND STRESS  *****
C   ****************************************************
```

```
C -- CALCULATE TOTAL INERTIAL FORCES OF COMPONENTS (lbf)
C  RIM
      FRIM = 2*PI*RHOAL*OMEGA(I)**2*LW*(ROL**3-RIL**3)/(3*12*GC)
C  SPOKES
      FSPOKE = SPOKEL*RHOAL*TSL*OMEGA(I)**2*(A/3*(RIL**3-RHL**3)+
     &         B/2*(RIL**2-RHL**2))/(12*GC)
C  HUB
      FHUB = 2*PI*RHOAL*OMEGA(I)**2*(A1/4*(RHL**4-RSH**4)+
     &       B1/3*(RHL**3-RSH**3))/(12*GC)

      R = RSH

C -- CALCULATE INERTIAL FORCE ACTING ON AN AREA OF ELEMENT
 600  IF (R .LE. RHL) THEN
C EQUATION FOR TOTAL INERTIAL FORCE ON HUB ELEMENT
          FI = FRIM+FSPOKE+FHUB - 2*PI*RHOAL*OMEGA(I)**2*
     &         (A1/4*(R**4-RSH**4)+B1/3*(R**3-RSH**3))/
     &         (12*GC)
C AREA OF HUB ELEMENT
          AI = 2*PI*R*(A1*R+B1)
      ELSE
          IF (R .LE. RIL) THEN
C EQUATION FOR TOTSL INERTIAL FORCE ON SPOKES
              FI = FRIM+FSPOKE - SPOKEL*RHOAL*TSL*OMEGA(I)**2*
     &             (A/3*(R**3-RHL**3)+B/2*(R**2-RHL**2))/(12*GC)
C AREA OF SPOKE ELEMENT
              AI = SPOKEL*TSL*(A*R + B)
          ELSE
C EQUATION FOR TOTAL INERTIAL FORCE ON RIM
              FI = FRIM - 2*PI*RHOAL*OMEGA(I)**2*LW*
     &             (R**3-RIL**3)/(3*12*GC)
C AREA OF RIM ELEMENT
              AI = 2*PI*R*LW
          ENDIF
      ENDIF

      DDEFI(I) = FI*DELR/(AI*E)
      RDEFI(I) = DDEFI(I)/2
      DEF(I) = DEF(I) + RDEFI(I)

      RADIAL(I) = FI/AI
      IF (RADIAL(I) .GT. MXSTRS(I)) THEN
          MXSTRS(I) = RADIAL(I)
      ENDIF

C   *******************************************************
C   *****  WRITE RADIAL STRESS AND DEFLECTION TO FILE  *****
C   *******************************************************
      WRITE(K,235)R,DEF(I),RADIAL(I)
 235  FORMAT(2X,F7.3,3X,F8.6,3X,F8.2)

      IF (R .LT. ROL) THEN
          R = R + DELR
          GOTO 600
      ENDIF
```

```
C  *******************************************
C  *****  SPEED-UP-TIME DEPENDANT LOOP  *****
C  *******************************************
       DO 520 J=1,3
       TORQUE(I,J) = TOTI*OMEGA(I)/(RUNUP(J)*12*GC)
       POWER(I,J) = TORQUE(I,J)*OMEGA(I)/6600
 520   FIBER(I,J) = ROL*OMEGA(I)*RUNUP(J)/2*(.3048/(12*1000))

 510   CONTINUE

C  **************************************************************
C  *****  PRINT SYSTEM REQUIREMENTS AND OTHER INFORMATION  *****
C  **************************************************************
       WRITE(4,240)RPMCRT
 240   FORMAT(/7X,'SYSTEM CRITICAL SPEED = ',F8.1,' RPM')

       WRITE(4,250)
 250   FORMAT(
      &/26X,'MAXIMUM POWER AND TORQUE REQUIRED',
      & /20X,'AND AMOUNT OF FIBER PAYED-OUT DURING SPEED-UP',//,
      & 33X,'PAYOUT VELOCITY')

       WRITE(4,260) MACH(1),MACH(2),MACH(3)
 260   FORMAT(16X,3(7X,'MACH',F4.1,5X))

       WRITE(4,270) RPM(1),RPM(2),RPM(3)
 270   FORMAT(16X,3(4X,'(',F7.1,' RPM)',3X))

       WRITE(4,280)
 280   FORMAT(7X,'SPEED-UP TIME',/11X,'(sec)')

       DO 540 I = 1,3

       WRITE(4,290) RUNUP(I),POWER(1,I),POWER(2,I),POWER(3,I)
 290   FORMAT(11X,F4.1,4X,3(3X,F6.1,' HP',8X))

       WRITE(4,300) TORQUE(1,I),TORQUE(2,I),TORQUE(3,I)
 300   FORMAT(18X,3(2X,F7.1,' in-lbf',4X))

       WRITE(4,310) FIBER(1,I),FIBER(2,I),FIBER(3,I)
 310   FORMAT(18X,3(6X,F4.2,' km',7X),/)
 540   CONTINUE

       WRITE(4,320) DEF(1),DEF(2),DEF(3)
 320   FORMAT(9X,'RADIAL',/7X,'DEFLECTION',3X,3(3X,F6.5,' in',8X))

       WRITE(4,330) MXSTRS(1),MXSTRS(2),MXSTRS(3)
 330   FORMAT(/7X,'MAX RADIAL',/9X,'STRESS',5X,3(2X,F7.0,' psi',7X))

       END
```

APPENDIX C

For 25" Drum

$L_{WH} = 2.33$ in

$\theta_{SP} = 12.0$ deg

Mass of Inertial Section:

$$m = \rho\, \theta_{SP}\, L_{WH}\, t_R\, R_m \qquad R_m = \frac{R_{IL} + R_{OL}}{2} = 12.25 \text{ in}$$

$$m = (0.098 \text{ lbm/in}^3)(12 \text{ deg}) \frac{\pi \text{ rad}}{180 \text{ deg}} (2.33 \text{ in})(0.5 \text{ in})(12.25 \text{ in})$$

$$m = 0.293 \text{ lbm}$$

Inertial Force of Element:

$$F = \frac{m R_{OL} \omega^2}{g_c}$$

$$F = \frac{(0.293 \text{ lb}_m)(12.5 \text{ in})(1075.2 \text{ rad/s})^2}{32.174 \frac{\text{lbm}}{\text{slug}} \quad 12 \frac{\text{in}}{\text{ft}}}$$

$$F = 10967 \text{ lbf}$$

Now this force must be distributed over proper length for stress calculations.

Looking between two spoke wheels



$$w = \frac{F}{L_{WH}} = 4707 \text{ lbf/in}$$

$$M_{MAX} = \frac{w L_{WH}^2}{12} = \frac{(4707 \text{ lbf/in})(2.33 \text{ in})^2}{12}$$

$$M_{MAX} = 2129 \text{ in-lbf}$$

$$\sigma_B = \frac{Mc}{I} = \frac{M \frac{t_R}{2}}{\frac{1}{12} R_{IL} \theta_{SP} t_R^3}$$

$$\sigma_B = \frac{6\, M_{MAX}}{R\, \theta_{SP}\, t_R^2} = \frac{6\,(2129 \text{ in-lbf})}{(12 \text{ in})\,(12 \text{ deg})\, \frac{\pi \text{ rad}}{180 \text{ deg}}\, (.5 \text{ in}^2)}$$

$\sigma_B = 20330$ psi

Looking between two spokes



$$w = \frac{F}{R_{IL}\theta_{SP}}$$

$$w = \frac{10967 \text{ lbf}}{(12 \text{ in})(12 \text{ deg}) \ \frac{\pi \text{ rad}}{180 \text{ deg}}}$$

$$w = 4364 \text{ lbf/in}$$

$$M_{MAX} = \frac{w(R_{oL}\ \theta_{SP})^2}{12}$$

$$M_{MAX} = \frac{(4364 \text{ lbf/in})\left[(12.5 \text{ in})(12 \text{ deg}) \ \pi \frac{\text{rad}}{180 \text{ deg}}\right]^2}{12}$$

$$M_{MAX} = 2493 \text{ in-lbf}$$

$$\sigma_B = \frac{Mc}{I} = \frac{M \frac{t_R}{2}}{\frac{1}{12} L_{WH}\ t_R^3} = \frac{6\ M_{MAX}}{L_{WH}\ t_R^2}$$

$$\sigma_B = \frac{6(2493 \text{ in-lbf})}{(2.33 \text{ in})(.5 \text{ in})^2}$$

$$\sigma_B = 25679 \text{ psi}$$

DISTRIBUTION

	Copies
U. S. Army Materiel System Analysis Activity ATTN: AMXSY-MP Aberdeen Proving Ground, MD 21005	1
IIT Research Institute ATTN: GACIAC 10 W. 35th Street Chicago, IL 60616	1
AMSMI-RD, Dr. McCorkle Dr. Rhoades Dr. Stephens	1
AMSMI-RD-GC-T, Dr. Jacobs	1
AMSMI-RD-GC-L, Dr. Ruffin	1
Ms. Overstreet	1
Mr. Hester	1
AMSMI-RD-GC-C, Mr. Warren	1
Mr. Cayson	5
AMSMI-RD-CS-R	15
AMSMI-RD-CS-T	1
AMSMI-GC-IP, Mr. Bush	1